

МИНОБРНАУКИ РОССИИ  
Федеральное государственное бюджетное образовательное учреждение  
высшего образования

**«САРАТОВСКИЙ НАЦИОНАЛЬНЫЙ ИССЛЕДОВАТЕЛЬСКИЙ  
ГОСУДАРСТВЕННЫЙ УНИВЕРСИТЕТ ИМЕНИ Н. Г.  
ЧЕРНЫШЕВСКОГО»**

**ЛАБОРАТОРНАЯ РАБОТА №5**

Отчёт о практике

студента 2 курса 251 группы  
направления 09.03.04 — Программная инженерия  
факультета КНИТ  
Тюменцева Радомира Александровича

Проверено:

Старший преподаватель

---

Е. М. Черноусова

Саратов 2025

## 1 Задание 5

**Вариант 1.** Изображение показано на рисунке 5.4. и состоит из 6 строк символов начиная с символа A (ASCII 41h) и далее по алфавиту с разными атрибутами начиная с 09h и далее плюс один. В каждой строке количество символов увеличивается на 1, начальная позиция вывода 5:10. Надо выполнить задание, используя прямую работу с видеопамятью.



Рис. 1 – Рисунок 5.4

### 1.1 Алгоритм программы

Программа настраивает доступ к видеопамяти через сегмент ES, сохраняет текущий видеорежим и переключает экран в текстовый режим 03h с активной страницей 0.

Процедура B10DISPLAY выводит в видеопамять ступенчатый набор символов. Начинается вывод с позиции строки 5 и столбца 10, используя символ A и атрибут 09h. Адрес видеопамяти для каждой строки вычисляется по формуле ((строка 80) + столбец) 2.

Внутренний цикл записывает в видеопамять несколько одинаковых символов, количество которых уменьшается по мере прохода внешнего цикла. После каждой строки символ и его атрибут увеличиваются, а номер строки сдвигается вниз.

После вывода всех строк программа ждёт нажатия клавиши, затем восстанавливает исходный видеорежим и завершает работу.

### 1.2 Текст программы на языке ассемблера с комментариями

```
1 .model small  
2 .stack 100h  
3 .186
```

```

4 .code
5 start:
6     mov AX, @data
7     mov DS, AX
8
9     ; Настраиваем запись данных в видеопамять на страницу 0
10    mov AX, 0B800h
11    mov ES, AX
12
13    mov AH, 0Fh
14    int 10h
15    push AX; Сохранение текущего видеорежима и активной страницы
16
17    ; Установка видеорежима 03h (80x25, 16 цветов)
18    mov AH, 00h
19    mov AL, 03h
20    int 10h
21
22    ; Установка активной страницы 0
23    mov AH, 05h
24    mov AL, 00h
25    int 10h
26
27    call B10DISPLAY
28
29    ; Ожидание нажатия любой клавиши
30    mov AH, 00h
31    int 16h
32
33    ; Восстановление исходного видеорежима и страницы
34    pop AX
35    mov AH, 00h
36    int 10h
37
38    ; Завершение программы
39    mov AX, 4C00h
40    int 21h
41
42    ; Процедура вывода изображения в видеопамять
43    B10DISPLAY proc
44        pusha
45
46        ; Начальные координаты:
47        mov CX, 5; Стока
48        mov BP, 10; Столбец
49
50        mov AL, 41h; ASCII-код символа А
51        mov AH, 09h; Атрибут светло-синего цвета
52
53        mov SI, 6; Счётчик строк

```

```

54 rows_loop:
55 ; Вычисление адреса в видеопамяти для текущей позиции
56 ; Адрес = ((строка * 80) + столбец) * 2
57 push AX
58 mov AX, CX      ; AX = строка
59 mov BX, 80
60 mul BX          ; AX = строка * 80
61 add AX, BP      ; AX = (строка * 80) + столбец
62 mov BX, 2
63 mul BX          ; AX = ((строка * 80) + столбец) * 2
64
65 mov DI, AX      ; DI = указатель на начало строки в видеопамяти
66 pop AX
67
68 ; Количество символов в текущей строке = 7 - SI
69 mov BX, 7
70 sub BX, SI
71 cols_loop:
72 mov ES:word ptr[DI], AX; Запись символа и атрибута в видеопамять
73 add DI, 2; Сдвиг на следующую позицию
74 dec BX
75 cmp BX, 0
76 jne cols_loop
77
78 ; Изменение символа и атрибута
79 inc AL
80 inc AH
81
82 ; Сдвиг на следующую строку
83 inc CX
84
85 ; Уменьшение счётчика строк
86 dec SI
87 cmp SI, 0
88 jne rows_loop
89
90 popa
91 ret
92 B10DISPLAY endp
93
94 end start

```

Текст программы

### 1.3 Скриншот запуска программы



Рис. 2 – Запуск программы

## **2 Ответы на контрольные вопросы**

### **1. Каков адрес области видеоданных для**

- режимов **00h - 06h**;
- **монохромного текстового режима**;

Для цветных текстовых и графических режимов **00h - 06h** начальный адрес равен **B800h**.

Для монохромного текстового режима (**07h**) начальный адрес равен **B000h**.

### **2. Укажите число страниц, разрешение и число цветов для видеорежима 03.**

- Разрешение: 80 столбцов на 25 строк, 720x400 пикселей
- Число цветов символа: 16
- Число страниц: 4

### **3. Укажите в двоичной форме содержимое байтов атрибутов для**

- **пурпурных символов на голубом фоне**;
- **зеленых символов на белом мигающем фоне**.

	<b>BL</b>	<b>R</b>	<b>G</b>	<b>B</b>	<b>I</b>	<b>R</b>	<b>G</b>	<b>B</b>
<b>Бит</b>	7	6	5	4	3	2	1	0
<b>A</b>	0	0	1	1	0	1	0	1
<b>B</b>	1	1	1	1	0	0	1	0

### **4. Объясните, как ограничивается количество доступных цветов для символа и для фона структурой байта атрибутов.**

Байт атрибутов определяет свойства каждого выводимого символа. Когда программа устанавливает атрибут, он остается в установленном состоянии до следующего явного изменения.

	<b>Фон</b>				<b>Символ</b>			
<b>Атрибут</b>	<b>BL</b>	<b>R</b>	<b>G</b>	<b>B</b>	<b>I</b>	<b>R</b>	<b>G</b>	<b>B</b>
<b>Номер бита</b>	7	6	5	4	3	2	1	0

Буквы R, G, B указывают позиции битов, соответствующих красному, зеленому, и синему цветам.

- Бит 7 (BL) устанавливает атрибут мерцания (может быть заблокирован)
- Биты 6-4 определяют цвет фона символа
- Бит 3 (I) устанавливает для символа нормальную (0) или повышенную (1) яркость.
- Биты 2-0 определяют составляющие цвета символа.

Фон может иметь один из восьми цветов, а сам символ один из шестнадцати:

Цвет	I	R	G	B	Шестнадцатеричный код
Черный	0	0	0	0	0
Синий	0	0	0	1	1
Зеленый	0	0	1	0	2
Голубой	0	0	1	1	3
Красный	0	1	0	0	4
Пурпурный	0	1	0	1	5
Коричневый	0	1	1	0	6
Белый	0	1	1	1	7
Серый	1	0	0	0	8
Светло-синий	1	0	0	1	9
Светло-зеленый	1	0	1	0	A
Светло-голубой	1	0	1	1	B
Светло-красный	1	1	0	0	C
Светло-пурпурный	1	1	0	1	D
Желтый	1	1	1	0	E
Ярко-белый	1	1	1	1	F

Рис. 3 – Таблица типичных значений атрибутов символов.

## 5. Укажите инструкции, необходимые для вывода на экран с помощью функции 09h прерывания INT 10h

- 10 желтых сердечек (ASCII 03h) на синем фоне;
- 5 белых звездочек (ASCII 2Ah) на красном фоне.
- 10 желтых сердечек (ASCII 03h) на синем фоне:

```

1 mov AH, 09h ; Запросить вывод (в текстовом режиме)
2 mov AL, 03h ; Символ сердечка
3 mov BH, 00h ; Страница 0
4 mov BL, 1Eh ; Синий фон, желтый символ
5 mov CX, 10 ; Число выводимых символов
6 int 10h      ; Вызвать обработчик прерывания

```

- 5 белых звёздочек (ASCII 2Ah) на красном фоне:

```

1 mov AH, 09h ; Запросить вывод (в текстовом режиме)
2 mov AL, 2Ah ; Символ звёздочки
3 mov BH, 00h ; Страница 0
4 mov BL, 47h ; Красный фон, белый символ
5 mov CX, 5   ; Число выводимых символов
6 int 10h      ; Вызвать обработчик прерывания

```